NAG Fortran Library Routine Document

F08KPF (ZGESVD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08KPF (ZGESVD) computes the singular value decomposition (SVD) of a complex m by n matrix A, optionally computing the left and/or right singular vectors.

2 Specification

```
      SUBROUTINE F08KPF (JOBU, JOBVT, M, N, A, LDA, S, U, LDU, VT, LDVT, WORK,
     1                   LWORK, RWORK, INFO)
      INTEGER            M, N, LDA, LDU, LDVT, LWORK, INFO
      double precision   S(*), RWORK(*)
      complex*16         A(LDA,*), U(LDU,*), VT(LDVT,*), WORK(*)
      CHARACTER*1        JOBU, JOBVT
```

The routine may be called by its LAPACK name ***zgesvd***.

3 Description

The SVD is written as

$$A = U\Sigma V^H,$$

where Σ is an m by n matrix which is zero except for its $\min(m,n)$ diagonal elements, U is an m by m unitary matrix, and V is an n by n unitary matrix. The diagonal elements of Σ are the singular values of A; they are real and non-negative, and are returned in descending order. The first $\min(m,n)$ columns of U and V are the left and right singular vectors of A.

Note that the routine returns V^H, not V.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBU – CHARACTER*1 *Input*

On entry: specifies options for computing all or part of the matrix U:

if JOBU = 'A', all m columns of U are returned in array U;
if JOBU = 'S', the first $\min(m,n)$ columns of U (the left singular vectors) are returned in the array U;
if JOBU = 'O', the first $\min(m,n)$ columns of U (the left singular vectors) are overwritten on the array A;
if JOBU = 'N', no columns of U (no left singular vectors) are computed.

Constraint: JOBU = 'A', 'S', 'O' or 'N'.

2: JOBVT – CHARACTER*1 *Input*

On entry: specifies options for computing all or part of the matrix V^H:

if JOBVT = 'A', all n rows of V^H are returned in the array VT;
if JOBVT = 'S', the first $\min(m, n)$ rows of V^H (the right singular vectors) are returned in the array VT;
if JOBVT = 'O', the first $\min(m, n)$ rows of V^H (the right singular vectors) are overwritten on the array A;
if JOBVT = 'N', no rows of V^H (no right singular vectors) are computed.

JOBVT and JOBU cannot both be 'O'.

3: M – INTEGER *Input*

On entry: m, the number of rows of the input matrix A.

Constraint: $\mathrm{M} \geq 0$.

4: N – INTEGER *Input*

On entry: n, the number of columns of the input matrix A.

Constraint: $\mathrm{N} \geq 0$.

5: A(LDA,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the m by n matrix A.

On exit: if JOBU = 'O', A is overwritten with the first $\min(m, n)$ columns of U (the left singular vectors, stored columnwise).

If JOBVT = 'O', A is overwritten with the first $\min(m, n)$ rows of V^H (the right singular vectors, stored rowwise).

If JOBU $\neq$ 'O' and JOBVT $\neq$ 'O', the contents of A are destroyed.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08KPF (ZGESVD) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{M})$.

7: S($*$) – ***double precision*** array *Output*

Note: the dimension of the array S must be at least $\max(1, \min(\mathrm{M}, \mathrm{N}))$.

On exit: the singular values of A, sorted so that $\mathrm{S}(i) \geq \mathrm{S}(i+1)$.

8: U(LDU,$*$) – ***complex*16*** array *Output*

Note: the second dimension of the array U must be at least $\max(1, ucol)$, where $ucol$ is the number of columns of U requested.

On exit: $(\mathrm{LDU}, \mathrm{M})$ if JOBU = 'A' or $(\mathrm{LDU}, \min(\mathrm{M}, \mathrm{N}))$ if JOBU = 'S'.

If JOBU = 'A', U contains the m by m unitary matrix U.

If JOBU = 'S', U contains the first $\min(m, n)$ columns of U (the left singular vectors, stored columnwise).

If JOBU = 'N' or 'O', U is not referenced.

9: LDU – INTEGER *Input*

On entry: the first dimension of the array U as declared in the (sub)program from which F08KPF (ZGESVD) is called.

Constraints:

if JOBU $=$ 'S' or 'A', LDU $\geq \max(1, \mathrm{N})$;
LDU ≥ 1 otherwise.

10: VT(LDVT,$*$) – ***complex*16*** array *Output*

Note: the second dimension of the array VT must be at least $\max(1, \mathrm{N})$.

On exit: if JOBVT $=$ 'A', VT contains the n by n unitary matrix V^H.

If JOBVT $=$ 'S', VT contains the first $\min(m, n)$ rows of V^H (the right singular vectors, stored rowwise).

If JOBVT $=$ 'N' or 'O', VT is not referenced.

11: LDVT – INTEGER *Input*

On entry: the first dimension of the array VT as declared in the (sub)program from which F08KPF (ZGESVD) is called.

Constraints:

if JOBVT $=$ 'A', LDVT $\geq \max(1, \mathrm{N})$;
if JOBVT $=$ 'S', LDVT $\geq \max(1, \min(\mathrm{M}, \mathrm{N}))$;
LDVT ≥ 1 otherwise.

12: WORK($*$) – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO $= 0$, WORK(1) returns the optimal LWORK.

13: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08KPF (ZGESVD) is called.

For good performance, LWORK should generally be larger. Consider increasing LWORK by at least $nb \times \min(\mathrm{M}, \mathrm{N})$, where nb is the optimal block size.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Constraint: LWORK $\geq \max(1, 2 \times \min(\mathrm{M}, \mathrm{N}) + \max(\mathrm{M}, \mathrm{N}))$.

14: RWORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, 5 \times \min(\mathrm{M}, \mathrm{N}))$.

On exit: if INFO > 0, RWORK$(1 : \min(\mathrm{M}, \mathrm{N}) - 1)$ contains the unconverged super-diagonal elements of an upper bidiagonal matrix B whose diagonal is in S (not necessarily sorted). B satisfies $A = UBV^H$, so it has the same singular values as A, and singular vectors related by U and V^H.

15: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value.

INFO > 0

If F08KPF (ZGESVD) did not converge, INFO specifies how many super-diagonals of an intermediate bidiagonal form did not converge to zero. See the description of RWORK above for details.

7 Accuracy

The computed singular value decomposition is nearly the exact singular value decomposition for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. In addition, the computed singular vectors are nearly orthogonal to working precision. See Section 4.9 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating point operations is approximately proportional to mn^2 when $m > n$ and m^2n otherwise.

The singular values are returned in descending order.

The real analogue of this routine is F08KBF (DGESVD).

9 Example

To find the singular values and left and right singular vectors of the 6 by 4 matrix

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i & -0.05 + 0.41i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i & -0.81 + 0.56i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i & -1.11 + 0.60i \\ -0.37 + 0.38i & 0.19 - 0.54i & -0.98 - 0.36i & 0.22 - 0.20i \\ 0.83 + 0.51i & 0.20 + 0.01i & -0.17 - 0.46i & 1.47 + 1.59i \\ 1.08 - 0.28i & 0.20 - 0.12i & -0.07 + 1.23i & 0.26 + 0.26i \end{pmatrix},$$

together with approximate error bounds for the computed singular values and vectors.

The example program for F08KRF (ZGESDD) illustrates finding a singular value decomposition for the case $m \leq n$.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08KPF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NB, NMAX
      PARAMETER        (MMAX=10,NB=64,NMAX=8)
      INTEGER          LDA, LDVT, LWORK
      PARAMETER        (LDA=MMAX,LDVT=NMAX,
     +                 LWORK=MMAX+3*NMAX+NB*(MMAX+NMAX))
```

```
*     .. Local Scalars ..
      DOUBLE PRECISION EPS, SERRBD
      INTEGER          I, IFAIL, INFO, J, LWKOPT, M, N
*     .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), DUMMY(1,1), VT(LDVT,NMAX),
     +                 WORK(LWORK)
      DOUBLE PRECISION RCONDU(NMAX), RCONDV(NMAX), RWORK(5*NMAX),
     +                 S(NMAX), UERRBD(NMAX), VERRBD(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         DDISNA, X04DAF, ZGESVD
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08KPF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*
*        Read the m by n matrix A from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
*
*        Compute the singular values and left and right singular vectors
*        of A (A = U*S*(V**H), m.ge.n)
*
         CALL ZGESVD('Overwrite A by U','Singular vectors (V)',M,N,A,
     +               LDA,S,DUMMY,1,VT,LDVT,WORK,LWORK,RWORK,INFO)
         LWKOPT = WORK(1)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Singular values'
            WRITE (NOUT,99999) (S(J),J=1,N)
*
            IFAIL = 0
            CALL X04DAF('General',' ',M,N,A,LDA,
     +                  'Left singular vectors (first n columns of U)',
     +                  IFAIL)
            WRITE (NOUT,*)
            CALL X04DAF('General',' ',N,N,VT,LDVT,'Conjugates of '//
     +                  'right singular vectors by row (V**H)',IFAIL)
*
*           Get the machine precision, EPS and compute the approximate
*           error bound for the computed singular values.  Note that for
*           the 2-norm, S(1) = norm(A)
*
            EPS = X02AJF()
            SERRBD = EPS*S(1)
*
*           Call DDISNA (F08FLF) to estimate reciprocal condition
*           numbers for the singular vectors
*
            CALL DDISNA('Left',M,N,S,RCONDU,INFO)
            CALL DDISNA('Right',M,N,S,RCONDV,INFO)
*
*           Compute the error estimates for the singular vectors
*
            DO 20 I = 1, N
               UERRBD(I) = SERRBD/RCONDU(I)
               VERRBD(I) = SERRBD/RCONDV(I)
   20       CONTINUE
*
*           Print the approximate error bounds for the singular values
*           and vectors
*
            WRITE (NOUT,*)
```

```
            WRITE (NOUT,*) 'Error estimate for the singular values'
            WRITE (NOUT,99998) SERRBD
            WRITE (NOUT,*)
            WRITE (NOUT,*)
     +        'Error estimates for the left singular vectors'
            WRITE (NOUT,99998) (UERRBD(I),I=1,N)
            WRITE (NOUT,*)
            WRITE (NOUT,*)
     +        'Error estimates for the right singular vectors'
            WRITE (NOUT,99998) (VERRBD(I),I=1,N)
         ELSE
            WRITE (NOUT,99997) 'Failure in ZGESVD. INFO =', INFO
         END IF
*
*        Print workspace information
*
         IF (LWORK.LT.LWKOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99996) 'Optimum workspace required = ', LWKOPT,
     +        'Workspace provided         = ', LWORK
         END IF
      ELSE
         WRITE (NOUT,*) 'MMAX and/or NMAX too small'
      END IF
      STOP
*
99999 FORMAT (3X,(8F8.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4)
99996 FORMAT (1X,A,I5,/1X,A,I5)
      END
```

9.2 Program Data

```
F08KPF Example Program Data

   6            4                                       :Values of M and N

 ( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06) (-0.05, 0.41)
 (-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42) (-0.81, 0.56)
 ( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17) (-1.11, 0.60)
 (-0.37, 0.38) ( 0.19,-0.54) (-0.98,-0.36) ( 0.22,-0.20)
 ( 0.83, 0.51) ( 0.20, 0.01) (-0.17,-0.46) ( 1.47, 1.59)
 ( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23) ( 0.26, 0.26) :End of matrix A
```

9.3 Program Results

```
 F08KPF Example Program Results

 Singular values
     3.9994  3.0003  1.9944  0.9995
 Left singular vectors (first n columns of U)
          1       2       3       4
 1  -0.5634 -0.2687  0.2451  0.3787
     0.0016 -0.2749  0.4657  0.2987

 2   0.1205 -0.2909  0.4329 -0.0182
    -0.6108  0.1085 -0.1758 -0.0437

 3  -0.0816 -0.1660 -0.4667 -0.0800
     0.1613  0.3885  0.3821 -0.2276

 4   0.1441  0.1984 -0.0034  0.2608
    -0.1532 -0.1737  0.1555 -0.5382

 5  -0.2487  0.6253  0.2643  0.1002
    -0.0926  0.3304 -0.0194  0.0140

 6  -0.3758 -0.0307  0.1266 -0.4175
     0.0793 -0.0816  0.1747 -0.4058
```

```
 Conjugates of right singular vectors by row (V**H)
          1       2       3       4
 1  -0.6971 -0.0867  0.0560 -0.1878
     0.0000 -0.3548 -0.5400 -0.2253

 2   0.2403  0.0725 -0.2477  0.7026
     0.0000 -0.2336 -0.5291  0.2177

 3  -0.5123 -0.3030  0.0678  0.4418
     0.0000 -0.1735  0.5162  0.3864

 4  -0.4403  0.5294 -0.3027  0.1667
     0.0000  0.6361 -0.0346  0.0258

 Error estimate for the singular values
        4.4E-16

 Error estimates for the left singular vectors
        4.4E-16     4.4E-16     4.5E-16     4.5E-16

 Error estimates for the right singular vectors
        4.4E-16     4.4E-16     4.5E-16     4.5E-16
```